Photo-Pair-Production Induced displacements of Atoms

A thesis submitted in partial fulfillment of the requirement
for the degree of Bachelor of Science with Honors in
Physics from the College of William and Mary in Virginia,

by

Loreto Peter Alonzi III.

Accepted for ______________________
Honors

____
Advisor: Dr. W.J. Kossler

____
Dr. Reilly

____
Dr. Orwoll

____
Dr. Carone

Williamsburg, Virginia
May 2005

Abstract

This paper presents a way to calculate the number of photo-pair-production displacements from a crystal lattice. A means to estimate the displacements was attempted using a technique by Jost, Luttinger, and Slotnick, however we show that this method is flawed. A second approach using the similarity to the bremsstrahlung interaction is given. The cross-section in terms of the momentum transfer, q, is then obtained by integrating that cross-section over the energies and angles and multiplied by a delta function in q. Screening is also accounted for. The Fortran program for this calculation is given.

Acknowledgements

Professor W.J. Kossler, The College of William and Mary

Professor T. Averett, The College of William and Mary

Contents

1 Introduction

Currently, NASA seeks a computer model that will approximate the damage space radiation inflicts on electronic components of a spacecraft. Space radiation contains many high energy neutrons, high energy atomic nuclei, electrons, positrons and photons. The flux of particles through the spacecraft's systems may damage them. These systems are usually made of semiconducting and insulating materials, such as Si and SiO_2. The molecular structure of these materials is, roughly, a lattice. The incident particles displace the atoms from their regular positions in the lattice. These displaced atoms can scatter and sometimes trap the electrons and holes which carry the current in the devices. Here we only consider the damage produced by photo-pair-production.

Examining Fig. 1 reveals that the cross-section for this interaction may be large.We expect the cross-section to be on the order of the Thompson cross-section or electronic Compton cross-section: $\pi r_0^2 = \pi e^2/m_e c^2$, much larger than the corresponding nuclear Compton cross-section: $\pi Z^2 e^2/M_{Nucleus} c^2$.



Figure 1: Feynman diagram for pair production in the Coulomb field of an atom.

To produce a displacement the energy transferred, $E = q^2/2M_{atom}$, must exceed a threshold energy which is on the order of $10eV$ [1]. To determine the number of displacements from a given flux several quantities must be evaluated: the density of the sample, the cross-section with respect to momentum transfer, and the number of displacements the initial displacement cause (primary-knock-ons). Previous work has established a function for the primary-knock-ons. The rest of the required information will be provided by NASA except for the calculation of the differential cross-section.

We will first present our implementation an elegant approach to obtaining this differential cross-section by Jost, Luttinger, and Slotnick[2]. Then, we show that the cross-section calculation method yields unphysical results and a new approach must be found. Next we present a solution stemming from work by Bethe and Heitler [3] that ultimately leads to a program to calculate $\frac{d\sigma}{dq}$. Then we account for screening by the electron cloud. In order to quickly calculate the cross-section we produced a program which is given in its entirety. We also included plots of several different parameters to graphically represent out results.

2 Total Number of Displacements

The following equation describes the total number of displacements a known flux of photons will cause in a sample due to pair-production.

$$\frac{dN}{dl} = \sum_i \int \frac{d\sigma_i}{dq}(E_\gamma, q) \cdot dq \cdot f_i(E_{pko}) \cdot \frac{dN}{dA}(Z_i) \cdot \Phi(E_\gamma). \tag{1}$$

N represents the total number of displacements and l is the length of the sample. The following components of Eq. 1 have already been determined. $f_i(E_{pko})$ is the total number of displacements that the primary-knock-ons will produce. $\Phi(E_\gamma)$ is the total flux incident on the sample. The only component of Eq. 1 left to be determined is $d\sigma/dq$.

3 Recoil Momentum Distribution from Unitarity

Jost, Luttinger, and Slotnick calculated the momentum transfer to the recoil nucleus in photo-pair-production covariantly by using the unitarity of the S matrix. They parameterized their equations in terms of $Q = q/2m$ and $\alpha = k_0/2m$. Using these parameters in their equation Jost arrived at [2]:

$$\sigma(k_0) = \int_{\alpha-\sqrt{\alpha^2-1}}^{\alpha+\sqrt{\alpha^2-1}} P(Q)dq \tag{2}$$

where,

$$P(Q, k_0) = \frac{Z^2 e^6}{32\pi^3 k_0^2 Q^3}[1 - F(Q^2)]^2 I(Q, \alpha) \tag{3}$$

and,

$$\begin{aligned} I(Q,\alpha) \;=\; & \left\{(1-2Q^2)J_1\right\} + \\ & \left\{(1 - 4Q^2 - 8Q\alpha + \frac{4Q^2-1}{3\alpha Q}) \cdot Ln[y^{\frac{1}{2}} + (y-1)^{\frac{1}{2}}]\right\} + \\ & \left\{(3 + \frac{2\alpha}{3Q} + \frac{2Q^2-1}{3\alpha Q})(y(y-1))^{\frac{1}{2}}\right\} + \\ & \left\{[-2(1+Q^2) + \frac{2\alpha^2}{3}(-4 + \frac{1}{Q^2})]\frac{1}{(1+1/Q^2)^{\frac{1}{2}}} \cdot \right. \\ & \left. Ln[\frac{(1+1/Q^2)^{1/2} - (1-1/y)^{1/2}}{(1+1/Q^2)^{1/2} + (1-1/y)^{1/2}}]\right\}. \end{aligned} \tag{4}$$

The braces, in Eq. 4, indicate portions of the equation that are graphed in Fig. 2. J_1 is defined in the following way,

$$J_1 = -R\left(\frac{1}{Z\lambda}\right) - \left(R\frac{\lambda}{Z} + \frac{\pi^2}{6} + \frac{ln\lambda}{2}\right) + \frac{(lnZ)^2}{2} - (lnZ(ln8Q\alpha)) \tag{5}$$

with,

$$\begin{aligned} R &= \int_0^t \frac{ln(1+x)}{x}dx, \\ Z &= [(y-1)^{1/2} + y^{1/2}]^2, \\ \lambda &= [Q + (Q^2+1)^{1/2}]^2, \\ y &= 2\alpha Q - Q^2. \end{aligned}$$



Figure 2: The line $I(Q, \alpha)$ is the plot of Equation 4, the other lines are components of the equation.

Figure 2 shows that the Jost, Luttinger, and Slotnick calculation for the cross-section goes negative at high Q, which is unphysical. To make sure that the plot was accurate the calculations were repeated. Again the result showed a negative cross-section. It appears that Jost, Luttinger, and Slotnick have made an error in their calculations and therefore the result cannot be relied upon.

4 The Bethe-Heitler Approach

4.1 Recoil Momentum Distributions by Change of Variable

In 1934 Bethe and Heitler [3], using perturbation theory, calculated cross-sections for bremsstrahlung and the related pair-production as functions of electron energy and various angles. In Heitler's book a comparison between the results of these calculations and experimental data [4].

$$
\begin{aligned}
d\sigma &= -\frac{Z^2}{137}\frac{e^4}{2\pi}\frac{p_+p_-dE_+}{k^3}\frac{\sin\theta_+\sin\theta_- d\theta_+ d\theta_- d\phi_+}{q^4}\cdot \\
&\quad [\frac{p_+^2\sin^2\theta_+}{(E_+-p_+\cos\theta_+)^2}(4E_-^2-q^2)\cdot \\
&\quad (4E_-^2-q^2)+\frac{p_-^2\sin^2\theta_-}{(E_--p_-\cos\theta_-)^2}(4E_+^2-q^2)+ \\
&\quad \frac{2p_+p_-\sin\theta_+\sin\theta_-\cos\phi_+}{(E_--p_-\cos\theta_-)(E_+-p_+\cos\theta_+)}(4E_+E_-+q^2-2k^2)- \\
&\quad 2k^2\frac{p_+^2\sin^2\theta_++p_-^2\sin^2\theta_-}{(E_--p_-\cos\theta_-)(E_+-p_+\cos\theta_+)}\Bigg], \qquad (6)
\end{aligned}
$$

with the following parameters:

$$
\begin{aligned}
k_0 &= E_+ + E_-, \\
E_+ &= -E, \\
E_- &= E_0, \\
p_+ &= p, \\
p_- &= p_0, \\
q^2 &= (k-p_+-p_-)^2.
\end{aligned}
$$

The angles Θ, Θ_0, and ϕ describe the direction of the electron in the initial and final state. Those angles are connected to the angles Θ_+, Θ_-, and ϕ_+ as follows [4]:

$$
\begin{aligned}
\Theta_+ &= \pi-\Theta, \\
\Theta_- &= \Theta_0, \\
\phi_+ &= \pi+\phi.
\end{aligned}
$$

Figure 3. shows a plot of Eq. 6.



Figure 3: Plot of $d\sigma/dE_+$ vs. E_+.

Eq. 6 determines the cross-section in terms of the positron energy and various angles. We can change from the variables E_+, Θ_+, Θ_-, and ϕ_+ to E_+, Θ_+, Θ_-, and q. Then solving for ϕ_+ as a function of q and taking $\frac{d\phi_+}{dq}|_{E_+,\Theta_+,\Theta_-}$ we arrive at Eq. 7.

$$d\phi_+ = -\frac{qdq}{p_+p_-sin\theta_+sin\theta_-sin\phi_+}. \tag{7}$$

Replacing $d\phi_+$ in Eq. 6 with Eq. 7 we come to:

$$\begin{aligned} d\sigma \quad = \quad & -\frac{Z^2}{137}\frac{e^4}{2\pi}\frac{dE_+}{k^3}\frac{d\theta_+d\theta_-qdq}{q^4\cdot\sin\phi_+}\cdot \\ & \left[\frac{p_+^2\sin^2\theta_+}{(E_+-p_+\cos\theta_+)^2}(4E_-^2-q^2)\cdot\right. \\ & (4E_-^2-q^2)+\frac{p_-^2sin^2\theta_-}{(E_--p_-\cos\theta_-)^2}(4E_+^2-q^2)+ \\ & \frac{2p_+p_-\sin\theta_+\sin\theta_-\cos\phi_+}{(E_--p_-\cos\theta_-)(E_+-p_+\cos\theta_+)}(4E_+E_-+q^2-2k^2)- \\ & \left.2k^2\frac{p_+^2\sin^2\theta_++p_-^2\sin^2\theta_-}{(E_--p_-\cos\theta_-)(E_+-p_+\cos\theta_+)}\right]. \end{aligned} \tag{8}$$

However this procedure leads to a new problem. There is now a $sin\phi_+$ in the denominator of Eq. 8. Presumably since the $d\sigma/dq$ is finite everywhere the numerator must also go to zero for $\phi_+\rightarrow 0$. However it is not clear in what manner the numerator approaches 0. Once evaluated Eq. 8 yields a $d\sigma/dq$ with unphysical fluctuations. If one were to average the fluctuations the result appears reasonable but this approach seems in elegant. This model provides a way to calculate cross-sections and momentum transfers but is not ideal. From here we backtracked to find another method to calculate the cross-section.

4.2 Using $\delta(q - q\prime)$

Since our previous attempt was not ideal we revert back to Eq. 6. For ease we will discuss Eq. 6 in the following terms:

$$\sigma = \int F(k_0, E_+, \Theta_+, \Theta_-, \phi_+) dE_+ d\Theta_+ d\Theta_- d\phi_+. \tag{9}$$

Our goal is to determine the cross-section as a function of momentum transfer. If one places a delta function inside the integral of Eq. 9 the result will be a calculation of $d\sigma/dq$. This equation then takes the form:

$$\frac{d\sigma}{dq} = \int F(k_0, E_+, \Theta_+, \Theta_-, \phi_+) \cdot \delta(q - q\prime) \cdot dE_+ d\Theta_+ d\Theta_- d\phi_+, \tag{10}$$

where

$$q\prime = \mid \vec{k_0} - (\vec{p_+} + \vec{p_-}) \mid . \tag{11}$$

Our goal is to produce a computer program to calculate the cross-section quickly. To achieve this we must use an approximation for the delta function in Eq. 10. A natural choice to approximate a delta function is a Gaussian with a small width. In the limit where the Gaussian has a small width the function will approximate a delta function.

$$\delta(q - q\prime) = (2\pi\Delta^2)^{-1/2} \lim_{\Delta \to 0} e^{-\frac{(q-q\prime)^2}{2\Delta^2}}. \tag{12}$$

Combining Eq. 10 and Eq. 12 we get:

$$\frac{d\sigma}{dq} = \frac{1}{\sqrt{(2\pi\Delta^2)}} \int e^{-\frac{(q-q\prime)^2}{2\Delta^2}} F(k_0, E_+, \Theta_+, \Theta_-, \phi_+) dE_+ d\Theta_+ d\Theta_- d\phi_+. \tag{13}$$

5 Screening

Up until this point screening by the electron cloud has been ignored. Roy and Reed examined whether the screening would affect the overall cross-section and momentum transfer. They produced a pair of equations for the total cross-section using $\hbar\omega$ as the incoming photon energy [5].

$$\sigma_{unscreened} = \frac{28}{9}\alpha(Zr_0)^2 \left[Ln(\frac{2\hbar\omega}{E_0}) - \frac{109}{42} - f(Z) \right]. \tag{14}$$

$$\sigma_{screened} = \frac{28}{9}\alpha(Zr_0)^2 \left[Ln(\frac{183}{Z^{\frac{1}{3}}}) - \frac{1}{42} - f(Z) \right]. \tag{15}$$

In some cases Eq. 15 is significantly different that Eq. 14. Therefore screening cannot be ignored. From Eq. 15 Roy and Reed formulate a way to calculate $d\sigma_{screened}$ in terms of E_+, the energy of the positron [6].

$$\begin{aligned} d\sigma_{screened} &= \frac{4\alpha(Zr_0)^2 dE_+}{(\hbar\omega)^3} \cdot \left[\left(E_+^2 + E_-^2 + \frac{2}{3}E_+E_- \right) \cdot \right. \\ & \left. \cdot \left[Ln\left(\frac{183}{Z^{1/3}}\right) - f(Z) \right] + \frac{1}{9}E_+E_- \right]. \end{aligned} \tag{16}$$

Figure 4 shows the functional relationship between $d\sigma$ and the electron energy. This work led to further research by Motz, Olsen, and Koch in 1969. They developed a function $F(q)$ to adjust for screening of the nucleus [7].



Figure 4: Plot of $d\sigma/dE_-$ from Equation 16.

$$F(q) = \left[1 + \left(111\frac{q}{m}Z^{1/3}\right)^2\right]^{-1} \tag{17}$$

They incorporated the correction into the Heitler form by multiplying Eq. 13 by $[1 - F(q)]^2$. This correction forces $\sigma \to 0$ as $q \to 0$ [7]. This result is expected since for $q = 0$ the charge is completely screened.

$$\frac{d\sigma}{dq} = \frac{[1 - F(q)]^2}{(2\pi\sigma^2)^{1/2}} \int e^{-\frac{(q-q\prime)^2}{2\Delta^2}} F(k_0, E_+, \Theta_+, \Theta_-, \phi_+) dE_+ d\Theta_+ d\Theta_- d\phi_+. \tag{18}$$

6 Fortran Program

Calculating this cross-section by hand would be rather tedious so we have written a computer program in Fortran to carry out the computations. The program is divided into five sections. The first section defines all of the parameters we use in the program. At the end of this section the program also reads in the two necessary inputs, Z and k_0. The next section calculates the screening factor as given in Eq. 17. Following this section the program calculates the un-normalized cross section using a function called from later in the program and Eq. 12. The delta function has a width $(2\Delta^2)$ written in the program as "2.*delsq". The value of this width was chosen so that Eq. 12 would be satisfied and so that the final output would not be series of spikes but rather a smooth curve. The next section normalizes the result from the previous section. The final section of the program is a function called "dsf". This function is Eq. 6. This part of the program calculates the function to be called earlier in the program.

All that is needed to run the program is the desired energy photon and the Z of the target. The computation is rather quick and takes about three seconds to calculate $d\sigma/dq$. The program puts the data in an output file named "outhei". The following graphs are of $Z = 14$ and $Z = 32$ with values of k_0 including $2, 4, 8, 12,$ and $20 MeV$.

Here is the program:

```
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
c                                                                          c
c Program to calculate the momentum transfer distribution produced by       c
c pair production. Screening is included. wjk and P.Alonzi 3/28/05         c
c                                                                          c
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
      implicit none
      real Z,r0,pref ! Z atomic number of target. r0 classical radius
      real k0! incoming photon energy-momentum
      real Ep,Em,pp,pm,q, dEp,dq! energies and momenta of pair, Ep step
c Units of E and p are MeV c=1
      real tp,tm,tf ! thetas and phi
      real dang! delta of angle
      real m! mass of electron
      real Emax ! maximum energy of electron
```

```
      real Qmin,Qmax! minimum and maximum momentum txf.
      real Pi
      integer iqmax,iq,idel,iqmx
      parameter(iqmax=50)
      real dse,dsq(0:iqmax) ! dsigma/dEp dsigma/dq
      real dsf ! function called
      integer i,n,ist,iend
      external dsf
      parameter(m=.511,Pi=3.14159265,r0=2.8e-13)
      real delq,delsq
      parameter (Qmax=5.,delq=.2,delsq=delq**2)
      real sige,sigd ! total csections
      real norm
      real qc(0:iqmax),fqc,sf(0:iqmax)! Charge in ch., form factor,
c                                      and screening factor
      sige=0;sigd=0              ! initialized



      Z=14 ! for Si default
      print*,'Z=',Z,'  =?'
      read*,Z

      dq=Qmax/iqmax
      idel=2*delq/dq
c Calculation of the screening factor sf(iq)
      do iq=0,iqmax
         dsq(iq)=0.
```

```
      qc(iq)=iq*dq
      fqc=1./(1+(111*(qc(iq)/m)*Z**(-.333))**2)
      sf(iq)=(1.-fqc)**2
   enddo

   n=0
   dang=Pi/40.
   print*,'k0(MeV)=?'
   read*,k0

c Calculate the Un-normalized differential cross-section with delta function.
   Emax=k0-m; dEp=(Emax-m)/40.
   print*,'Emax=',Emax,',  Qmax=',Qmax,' dq=',dq
   open(10,file='outhei')
   do Ep=m,Emax,dEp
      Em=k0-Ep;pp=sqrt(Ep**2-m**2);
      if(Em<m)then
         print*,'aha Em=',Em
         Em=m
      endif
         pm=sqrt(Em**2-m**2)
      do tp=dang,Pi-dang,dang
         do tm=dang,Pi-dang,dang
            do tf=dang/2.,Pi-dang/2.,dang
               dse=dsf(k0,Ep,Em,pp,pm,tp,tm,tf,q)
               sige=sige+dse
               iqmx=q/dq
```

```
               if(iqmx<=idel)then
                  ist=0
               else
                  ist=iqmx-idel
               endif
               if(iqmx+idel>=iqmax)then
                  iend=iqmax
               else
                  iend=iqmx+idel
               endif
               do i=ist,iend
                  dsq(i)=dsq(i)+dse*exp(-(qc(i)-q)**2/(2.*delsq))
               enddo
            enddo
         enddo
      enddo
   enddo

c Normalize the differential cross-section and output to "outhei".
   do i=0,iqmax
      sigd=sigd+dsq(i)
   enddo
   norm=sige/(sigd*dq)
   pref=Z**2*r0**2*m**2/(137.*2*Pi)*dEp*dang**3*norm
   sigd=0.

   do i=0,iqmax
```

```
         dsq(i)=-dsq(i)*pref*sf(i)
         sigd=sigd+dsq(i)
         write(10,*)qc(i),dsq(i)
      enddo

      close(10)
      print*, 'sige=',-pref*sige,' sigd=',sigd
      end


cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
c                                                                          c
c   Differential Cross-section from Heitler 3rd Ed. p 257 Eq. 6.           c
c                                                                          c
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
      function dsf(k,Ep,Em,pp,pm,tp,tm,tf,q)
      implicit none
      real cstp,sntp,cstm,sntm,cstf
      real dsf,k,Ep,Em,pp,pm,tp,tm,tf,q
      real at,bt,ct,dt,et ! pieces of the formula
      cstf=cos(tf);cstp=cos(tp);cstm=cos(tm)
      sntp=sin(tp);sntm=sin(tm)
      q=sqrt(k**2-2.*k*(pp*cstp+pm*cstm)+
     $     pp*pp+pm*pm+2.*pp*pm*(cstp*cstm+
     $     sntp*sntm*cstf))
      at=pp*pm/k**3*sntp*sntm/q**4
      bt=pp**2*sntp**2/(Ep-pp*cstp)**2*(4*Em**2-q**2)
```

```
 ct=pm**2*sntm**2/(Em-pm*cstm)**2*(4*Ep**2-q**2)
 dt=2*pp*pm*sntp*sntm*cstf/
$ ((Em-pm*cstm)*(Ep-pp*cstp))
$ *(4.*Ep*Em+q**2-2*k**2)
 et=-2*k**2*(pp**2*sntp**2+pm**2*sntm**2)/
$ ((Em-pm*cstm)*(Ep-pp*cstp)) !$
 dsf=at*(bt+ct+dt+et)
 return
 end
```



Figure 5: Differential cross-section as a function of momentum transfered to the nucleus for 2 MeV photons incident on Si.



Figure 6: Differential cross-section as a function of momentum transfered to the nucleus for 4 MeV photons incident on Si.



Figure 7: Differential cross-section as a function of momentum transfered to the nucleus for 8 MeV photons incident on Si.



Figure 8: Differential cross-section as a function of momentum transfered to the nucleus for 12 MeV photons incident on Si.



Figure 9: Differential cross-section as a function of momentum transfered to the nucleus for 20 MeV photons incident on Si.



Figure 10: Differential cross-section as a function of momentum transfered to the nucleus for 2 MeV photons incident on Ge.



Figure 11: Differential cross-section as a function of momentum transfered to the nucleus for 4 MeV photons incident on Ge.



Figure 12: Differential cross-section as a function of momentum transfered to the nucleus for 8 MeV photons incident on Ge.



Figure 13: Differential cross-section as a function of momentum transfered to the nucleus for 12 MeV photons incident on Ge.



Figure 14: Differential cross-section as a function of momentum transfered to the nucleus for 20 MeV photons incident on Ge.

7 Conclusion

We began by determining an equation to calculate the number of photo-pair-production displacements per length of a sample. All of the necessary components were known except for $d\sigma/dq$. We proceeded to determine $d\sigma/dq$. The first method we tried using Jost's work produced unphysical results. Therefore we moved on to Heitler's calculation of $d\sigma$. Using a delta function we manipulated his result to produce $d\sigma/dq$. We then produced a program to quickly calculate this value. With a way to determine the differential cross-section the total number of displacements can now be calculate.

References

[1] M. Nastasi, J. W. Mayer, and J. K. Hirvonen, *Ion-Solid Interactions: Fundamentals and Applications* (Cambridge University Press, Cambridge, 1996), No. 146.

[2] R. Jost, M. Luttinger, and M. Slotnick, Physics Review **80**, 189 (1950).

[3] H. A. Bethe and W. Heitler, Proc. Roayl Society **A146**, 83 (1934).

[4] W. Heitler, in *The Quantum Theory of Radiation*, edited by N. F. Mott, E. C. Bullard, and D. H. Wilkinson (Oxford, London, 1960).

[5] H. Davies, H. A. Bethe, and L. C. Maximon, Physical Review **93**, 788 (1954).

[6] R. Roy and R. D. Reed, *Interactions of Photons and Leptons with Matter* (Academic Press, New York, 1968).

[7] J. Motz, H. A. Olsen, and H. W. Koch, Rev. Mod. Phys. **41**, 581 (1969).